

# HSPICE Simulation Example

Jiahao Kang

```
* Tutorial: An Simulation Example using HSPICE
* By Jiahao Kang
* F=(A+B)+'+(CD)'
```

```
.TITLE Example

***** Load library *****
.INCLUDE '65nm_bulk.pm'

***** Define parameters *****
.PARAM VDD=1
+ LMOS =65n
+ WN   =65n
+ WP   ='2*WN'
.TEMP 70

***** Define stimuli *****
VA A 0 pulse 0 VDD 0.2n 0.2n 0.2n 1n 2n
VB B 0 pulse 0 VDD 0.2n 0.2n 0.2n 1n 2n
VC C 0 pulse 0 VDD 0.2n 0.2n 0.2n 1n 2n
VD D 0 pulse 0 VDD 0.2n 0.2n 0.2n 1n 2n
VDD VDD 0 DC VDD

***** Define global nodes for use in subcircuits *****
.GLOBAL VDD

***** Define subcircuits (modules) *****

.SUBCKT INV X Y
M1 Y X 0 0 NMOS L=LMOS W=WN
M2 Y X VDD VDD PMOS L=LMOS W=WP
.ENDS

.SUBCKT NAND2 A B Y
M1 1 B 0 0 NMOS L=LMOS W=WN
M2 Y A 1 0 NMOS L=LMOS W=WN
M3 Y A VDD VDD PMOS L=LMOS W=WP
M4 Y B VDD VDD PMOS L=LMOS W=WP
.ENDS

.SUBCKT NOR2 A B Y
M1 Y B 0 0 NMOS L=LMOS W=WN
M2 Y A 0 0 NMOS L=LMOS W=WN
M3 Y A 1 VDD PMOS L=LMOS W=WP
M4 1 B VDD VDD PMOS L=LMOS W=WP
.ENDS

***** Define main circuit *****

* F=(A+B)+'+(CD)'
```

```
X1 A B 1 NOR2
X2 C D 2 NAND2
X3 1 2 xF NOR2
X4 xF F INV
* Load capacitance
CL F 0 0.1p
```

First line in a SPICE deck is always a comment (with or without \*).

Define parameters with .param  
Can define parameter based on other parameters or expressions.

Use '+' to continue long lines on the proceeding line.

Here's how you can do arithmetic on your parameters.

First vdd is the voltage source.  
Second vdd is the node.  
Third vdd is the parameter

Subcircuits are SPICE's way of defining modules repeated in your design.

Instantiate modules like so.

```

***** Analysis Options *****
.TRAN 0.1p 2n
.MEAS TRAN avgpower AVG power FROM=0ns TO=2ns
.MEAS TRAN tdlay TRIG V(A) VAL='VDD/2' RISE=1
                  + TARG V(F) VAL='VDD/2' FALL=1

***** Alter the parameters and run again *****

.ALTER CASE 2: WP=WN
.PARAM WP = '1*WN'

.ALTER CASE 3: Increase Vdd by 10%
.PARAM VDD = 1.1

.ALTER CASE 4: Change load capacitance
CL F 0 0.01p

.ALTER CASE 5: Change temperature
.TEMP 70

.ALTER CASE 6: Change stimuli
VA A 0 DC 0
VD D 0 DC VDD

.END

```

Power measurements

Measure propagation delays accurately using the '.meas' statement. Outputs are written to .lis file.

Here we trigger when the voltage at node 'A' crosses vdd/2, and measure the time until the output crosses vdd/2

.ALTER statements allows us to modify the circuit and run again. They must be before the final .end statement. Note: ALTER blocks are incremental!!!